The Kenton Study

The Kenton Food Company wished to test four different package designs for a new breakfast cereal. Twenty stores, with approximately equal sales volumes, were selected as the experimental units. Each store was randomly assigned one of the package designs, with each package design assigned to five stores. A fire occurred in one store during the study period, so this store had to be dropped from the study. Hence, one of the designs was tested in only four stores. The stores were chosen to be comparable in location and sales volume. Other relevant conditions that could affect sales, such as price, amount and location of shelf space, and special promotional efforts, were kept the same for all of the stores in the experiment. Sales, in number of cases, were observed for the study period, and the results are recorded in Table 16.1. This study is a completely randomized design with package design as the single, four-level factor.

(*Applied Linear Statistical Models*, 5th ed., pp. 724-725, Kutner et al., 2005)

TABLE 16.1 **Number of Cases Sold by Stores for Each of Four Package Designs—Kenton Food Company Example.**

Package Design	Store (j)				
	1	2	3	4	5
i	Y_{i1}	Y_{i2}	Y_{i3}	Y_{i4}	Y_{i5}
1	11	17	16	14	15
2	12	10	15	19	11
3	23	20	18	17	
4	27	33	22	26	28
All designs					

```
%  cat kenton.data
1     11
1     17
1     16
1     14
1     15
2     12
2     10
2     15
2     19
2     11
3     23
3     20
3     18
3     17
4     27
4     33
4     22
4     26
4     28


/********************** kenton1.sas **************************/
options linesize=79 pagesize=100 noovp formdlim=' ';
title 'Kenton Oneway Example From Neter et al.';

proc format;
     value pakfmt 1 = '3Colour Cartoon'   2 = '3Col No Cartoon'
                  3 = '5Colour Cartoon'   4 = '5Col No Cartoon';

data food;
     infile 'kenton.data';
     input package sales;
     label package = 'Package Design'
           sales   = 'Number of Cases Sold';
     format package pakfmt.;

/* Basic one-way ANOVA -- well, not completely basic */
proc glm;
     class package;
     model sales = package;
     /* Multiple comparisons. Tukey should be best */
     lsmeans package / pdiff  adjust=tukey;
     /* Test some custom contrasts */
     contrast '3Colourvs5Colour' package 1  1 -1 -1;
     contrast 'Cartoon'          package 1 -1  1 -1;
     contrast 'CartoonDepends'   package 1 -1 -1  1;
     /* Test a COLLECTION of contrasts */
     contrast 'Overall F = 18.59'  package 1 -1  0  0,
                                   package 0  1 -1  0,
                                   package 0  0  1 -1;
     /* Get estimated value of a contrast along with a test (F=t-squared) */
     estimate '3Colourvs5Colour' package 1  1 -1 -1 / divisor = 2;

/* Scheffe critical value for a test of s contrasts is critval * (p-1)/s.
   For p=4 means and a single contrast, it's critval *  (4-1)/1 = 3 */

proc iml;
     title2 'Critical value for all possible 1-df Scheffe tests';
     numdf = 3;   /* p-1 = Numerator degrees of freedom for initial test */
     dendf = 15;  /* n-p = Denominator degrees of freedom for initial test */
     alpha = 0.05;
     critval = finv(1-alpha,numdf,dendf);  print critval;
     ScheffeCritval = critval*numdf; print ScheffeCritval;
```

Kenton Oneway Example From Kutner et al. 1

The GLM Procedure

Class Level Information

Class	Levels	Values
package	4	3Col No Cartoon 3Colour Cartoon 5Col No Cartoon 5Colour Cartoon

Number of Observations Read 19
Number of Observations Used 19

Kenton Oneway Example From Kutner et al. 2

The GLM Procedure

Dependent Variable: sales Number of Cases Sold

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	3	588.2210526	196.0736842	18.59	<.0001
Error	15	158.2000000	10.5466667		
Corrected Total	18	746.4210526			

R-Square	Coeff Var	Root MSE	sales Mean
0.788055	17.43042	3.247563	18.63158

Source	DF	Type I SS	Mean Square	F Value	Pr > F
package	3	588.2210526	196.0736842	18.59	<.0001

Source	DF	Type III SS	Mean Square	F Value	Pr > F
package	3	588.2210526	196.0736842	18.59	<.0001

The GLM Procedure
Least Squares Means
Adjustment for Multiple Comparisons: Tukey-Kramer

package	sales LSMEAN	LSMEAN Number
3Col No Cartoon	13.4000000	1
3Colour Cartoon	14.6000000	2
5Col No Cartoon	27.2000000	3
5Colour Cartoon	19.5000000	4

Least Squares Means for effect package
Pr > |t| for H0: LSMean(i)=LSMean(j)

Dependent Variable: sales

i/j	1	2	3	4
1		0.9353	<.0001	0.0583
2	0.9353		0.0001	0.1549
3	<.0001	0.0001		0.0142
4	0.0583	0.1549	0.0142	

The GLM Procedure

Dependent Variable: sales Number of Cases Sold

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
3Colourvs5Colour	1	411.4000000	411.4000000	39.01	<.0001
Cartoon	1	49.7058824	49.7058824	4.71	0.0464
CartoonDepends	1	93.1882353	93.1882353	8.84	0.0095
Overall F = 18.59	3	588.2210526	196.0736842	18.59	<.0001

Parameter	Estimate	Standard Error	t Value	Pr > |t|
3Colourvs5Colour	-9.35000000	1.49705266	-6.25	<.0001

Critical value for all possible 1-df Scheffe tests

critval
3.2873821

ScheffeCritval
9.8621463

Comparing adjusted p-values for Tukey, Bonferroni, Scheffé

Adjustment for Multiple Comparisons: Tukey-Kramer

package	sales LSMEAN	LSMEAN Number
3Col No Cartoon	13.4000000	1
3Colour Cartoon	14.6000000	2
5Col No Cartoon	27.2000000	3
5Colour Cartoon	19.5000000	4

Dependent Variable: sales

i/j	1	2	3	4
1		0.9353	<.0001	0.0583
2	0.9353		0.0001	0.1549
3	<.0001	0.0001		0.0142
4	0.0583	0.1549	0.0142	

Adjustment for Multiple Comparisons: Bonferroni

i/j	1	2	3	4
1		1.0000	<.0001	0.0808
2	1.0000		0.0001	0.2397
3	<.0001	0.0001		0.0180
4	0.0808	0.2397	0.0180	

Adjustment for Multiple Comparisons: Scheffe

i/j	1	2	3	4
1		0.9507	<.0001	0.0895
2	0.9507		0.0002	0.2125
3	<.0001	0.0002		0.0248
4	0.0895	0.2125	0.0248	